

## MATRIX SIMPLEX SOLUTION

### PURPOSE

Compute the simplex solution to a linear programming problem.

### DESCRIPTION

Linear programming problems are used to solve problems where a number (call it m) of limited resources are to be allocated among a number (call it n) of competing activities. The linear programming problem can be formulated as follows:

1. Let  $X_j$  represent the level of activity j (for  $j = 1$  to  $n$ ).
2. Let  $B_i$  represent the amount of resource i available (for  $i = 1$  to  $m$ ).
3. Let  $A_{ij}$  represent the amount of resource i consumed by activity j.
4. Let Z be the overall measure of effectiveness and let  $C_j$  be the increase in Z that would result from each unit increase in  $X_j$ .

This can be represented mathematically as

$$\text{Maximize } Z = C_1X_1 + C_2X_2 + \dots + C_nX_n$$

subject to the restrictions

$$A_{11}X_1 + A_{12}X_2 + \dots + A_{1n}X_n \leq B_1$$

$$A_{21}X_1 + A_{22}X_2 + \dots + A_{2n}X_n \leq B_2$$

...

$$A_{m1}X_1 + A_{m2}X_2 + \dots + A_{mn}X_n \leq B_m$$

The function being maximized is called the objective function and the series of equations are called the constraints. It is assumed that the A, B, and C parameters are known. The MATRIX SIMPLEX SOLUTION command solves the series of constraints for the  $X_j$ . The VECTOR DOT PRODUCT command is then used to find the value for the objective function (i.e., Z). The above model also assumes that the  $X_j$  are non-negative. It can also handle greater than inequalities in the constraint functions.

### SYNTAX

LET <var2> = MATRIX SIMPLEX SOLUTION <var1> <mat> <SUBSET/EXCEPT/FOR qualification>

where <var1> is a variable containing the coefficients for the objective function;

<mat> is a matrix containing the coefficients for the constraint equations. It has the following form:

$$\langle \text{mat} \rangle = [ A_{ij} \mid \text{tag} \mid B_i ]$$

where  $A_{ij}$  and  $B_i$  are as defined above and tag is a column that defines the type of inequality (-1 for  $\leq$ , 0 for  $=$ , 1 for  $\geq$ );

<var2> is a variable containing the values for  $X_j$  computed by the simplex solution;

and where the <SUBSET/EXCEPT/FOR qualification> is optional and rarely used in this context.

### EXAMPLES

```
LET X = MATRIX SIMPLEX SOLUTION F C
```

```
LET PROFIT = VECTOR DOT PRODUCT F X
```

### NOTE

Many linear programming problems involve a large number of constraints and activities. DATAPLOT's linear programming capability is only intended for small to moderate size problems. Specifically, the default limits for the maximum number of rows and columns for a single matrix is 750 and 100 respectively (i.e., the <mat> matrix is limited by this). Even though these limits can be increased by re-installing DATAPLOT, we would recommend using more specialized linear programming software for problems of that magnitude.

### DEFAULT

None

### SYNONYMS

None

### RELATED COMMANDS

MATRIX INVERSE	=	Compute a matrix inverse.
MATRIX SOLUTION	=	Solve a system of linear equations.
PRINCIPAL COMPONENTS	=	Compute the principal components of a matrix.

VECTOR DOT PRODUCT = Compute the vector dot product of two variables.

## REFERENCE

“Numerical Recipes: The Art of Scientific Computing (FORTRAN Version),” Press, Flannery, Teukolsky, and Vetterling, Cambridge University Press, 1989 (chapter 10).

“Operations Research,” Hiller and Lieberman, Holden-Day, 1974.

## APPLICATIONS

Linear Programming

## IMPLEMENTATION DATE

87/10

## PROGRAM

. PURPOSE--DETERMINE PRODUCTION MIX OF 2 OILS TO MAXIMIZE  
 . PROFITS. ANALYSIS TECHNIQUE--SIMPLEX METHOD IN LINEAR  
 . PROGRAMMING. (4 VARIABLES, 10 CONSTRAINTS)  
 . APPLICATION--PRODUCTION OPTIMIZATION  
 . SOURCE--BRONSON, RICHARD, THEORY AND PROBLEMS OF OPERATIONS  
 . RESEARCH (SCHAUM'S OUTLINE SERIES), MCGRAW-HILL, 1982, PAGES 6, 7, 8, 40, AND 41.  
 . TO FIND--4 PRODUCTIONS SETTINGS--  
 . X1 = NUMBER OF BARRELS OF DOMESTIC OIL TO BLEND INTO REGULAR GAS.  
 . X2 = NUMBER OF BARRELS OF FOREIGN OIL TO BLEND INTO REGULAR GAS.  
 . X3 = NUMBER OF BARRELS OF DOMESTIC OIL TO BLEND INTO PREMIUM GAS.  
 . X4 = NUMBER OF BARRELS OF FOREIGN OIL TO BLEND INTO PREMIUM GAS.  
 . CONSTRAINTS ARE BASED ON--  
 . COST (/BBL) --DOMESTIC = \$8 FOREIGN = \$14  
 . INVENTORY (BBL) --DOMESTIC = 40,000 FOREIGN = 60,000  
 . OCTANE RATING --DOMESTIC = 87 FOREIGN = 98  
 . VAPOR PRESSURE --DOMESTIC = 25 FOREIGN = 15  
 . REVENUE (/BBL) --REGULAR = \$12 PREMIUM = \$14  
 . MAX DEMAND (BBL/WK)-REGULAR = 100,000 PREMIUM = 20,000  
 . MIN OCTANE --REGULAR = 88 PREMIUM = 93  
 . MAX VAPOR PRESSURE--REGULAR = 23 PREMIUM = 23  
 . NOTE--A FEASIBLE SOLUTION IS 0, 50000, 0, 5000. (BUT  
 . WITH A LOSS OF \$155000). THE OPTIMAL SOLUTION IS  
 . 37727.3, 12272.7, 2272.7, 2727.3. (PROFIT OF \$125000)

```

DIMENSION 100 VARIABLES
.  DEFINE THE COEFFICIENTS OF THE OBJECTIVE FUNCTION
LET F = DATA 4 -3 6 -1
.  DEFINE THE CONSTRAINTS CONSISTING OF LEFT-SIDE
.  COEFFICIENTS CODED INEQUALITY (< AS -1, = AS 0,
.  > AS +1) AND RIGHT-SIDE LIMITS.
READ MATRIX C
1 1 0 0 -1 100000
0 0 1 1 -1 20000
1 0 1 0 -1 40000
0 1 0 1 -1 60000
1 -1 0 0 0 -1 0
0 0 6 -5 -1 0
2 -8 0 0 -1 0
0 0 2 -8 -1 0
1 1 0 0 +1 50000
0 0 1 1 +1 5000
END OF DATA
PRINT F C
.  GENERATE THE SIMPLEX SOLUTION
LET X = MATRIX SIMPLEX SOLUTION F C
LET PROFIT = VECTOR DOT PRODUCT F X
PRINT F X PROFIT
    
```

The following output is generated.

```

PARAMETERS AND CONSTANTS--
    PROFIT  --  0.1250000E+06
VARIABLES--F          X
    0.4000000E+01  0.3772727E+05
   -0.3000000E+01  0.1227273E+05
    0.6000000E+01  0.2272727E+04
   -0.1000000E+01  0.2727272E+04
    
```